Модель файловой системы 
1 Введение
Основная цель данного материала – дать пример использования языка моделирования AsmL для описания, анализа и последующего тестирования систем. В разделе 2 приведена модель базовых операций над файлами, в разделе 3 описан случайный сценарий работы файловой системы, в разделе 4 показано, как использовать эту модель для создания тестового комплекта для файловой системы, в разделе 5 содержится заключительная дискуссия по данному вопросу.
2 Модель файловой системы
Сначала мы описываем базовые типы, используемые в модели.

type FileId   = Integer // file identifier
type FileTime = Integer // time stamp assigned to a file
type FileName = String  // local file name including extension
enum FileType

  Directory
  Regular
2.1 Файл
Все параметры файла собраны в одном классе FileInfo. Поля fid и sort являются константами, все остальные могут изменяться в течение времени.
class FileInfo // model of a file
  fid  as FileId                  // file identifier
  sort as FileType                // directory or regular 
  var name     as FileName        // file name relative to the parent dir
  var parent   as FileId          // identifier of the parent directory
  var data     as String          // data piled up in the file (ignored here)
  var children as Set of FileId   // set of identfiers of children resourses
  var creationTime   as FileTime  // creation time stamp
  var lastUpdateTime as FileTime  // time stamp of last update 
2.2 Конструктор класса FileSystem 
В классе FileSystem, описывающем файловую систему, кроме множества файлов files, содержится ссылка на корневой каталог root и переменная-таймер now.

class FileSystem

  root as FileId               // (= 0) file identifier of the root directory
  var files as Set of FileInfo // files kept by the file system
  var now   as FileTime        // current time
  FileSystem() // constructor of the file system class
    root  = 0
    files = { new FileInfo(root, Directory, ".", 0, "", {}, 0, 0) }

    now   = 1
2.3 Вспомогательные функции 

Нам понадобится функция File, которая по значению fid возвращает ссылку на соответствующий файл. Функция Known проверяет, используется ли заданное значение fileId для какого-то из существующих файлов. Функция Now возвращает текущее значение таймера, после чего увеличивает его на 1.
class FileSystem

  File(fid as FileId) as FileInfo // search file record by FileId vaule
    require Known(fid)
    return the file | file in files where file.fid = fid

  Known(fid as FileId) as Boolean // 
    return fid in { file.fid | file in files } // the value is used
                                               // for some exisitng file
  Now() as FileTime  // current time
    now += 1
    return now

2.4 Файловые операции
Основные переходы в нашей модели соответствуют файловым операциям: создание, редактирование, переименование и удаление файла.
2.4.1 Создание файла
Прежде, чем создать новый файл, нужно проверить, существует ли родительская директория, а также, не совпадает ли имя создаваемого файла с именем уже существующего файла в этой директории.

class FileSystem  
  CreateEnabled(      // preconditions for creation of a file
    parent as FileId,

    name as String

    ) as Boolean

    return Known(parent)

      and then (File(parent).sort = Directory)

      and then name notin { File(child).name | child in File(parent).children }

Когда файл создается, мы добавляем соответствующий объект в множество files и изменяем свойство children родительской директории.
class FileSystem  
  Create(

    parent as FileId,

    name as String,

    data as String,

    sort as FileType

    ) as FileId
    require CreateEnabled(parent, name)

    let fid = min id | id in {1..1000} where not Known(id)

    let time = Now()

    add new FileInfo(fid, sort, name, parent, data, {}, time, time) to files
    File(parent).children += { fid }

    return fid
2.4.2 Редактирование файла
Единственным предусловием для этой операции является существование редактируемого файла.
class FileSystem

  UpdateEnabled(fid as FileId) as Boolean

    return Known(fid)
Этот переход сыстемы соответствует тому, что мы изменяем соответствующим образом поля data и lastUpdateTime. 
class FileSystem

  Update(

    fid as FileId,

    data as String

    )

    require UpdateEnabled(fid)

    File(fid).data := data
    File(fid).lastUpdateTime := Now()
2.4.3 Удаление файла
Прежде, чем удалять файл, нужно проверить, что файл существует, что это не корневой каталог, и, если это директория, что она пуста.
class FileSystem

  DeleteEnabled(

    fid as FileId
    ) as Boolean
    return Known(fid)
           and then fid ne root
           and then File(fid).children = {}

Мы удаляем соответствующий объект из множества files и изменяем соответсвующее свойство родительской директории.
  Delete(

    fid as FileId

    )
    require DeleteEnabled(fid)

    let parent = File(File(fid).parent)
    remove fid from parent.children
    remove File(fid) from files

2.4.4 Переименование/перенос файла
Эта операция имеет несколько предусловий: все участвующие файлы должны существовать, переносить можно только в директорию, новое имя файла не должно совпадать с именем уже существующего файла в этой директории. Кроме этого, директория, в которую осуществляется перенос, не должна находиться среди «потомков» переносимого файла, т.е. нельзя переносить директорию в свою под-директорию.
class FileSystem

  MoveEnabled(

    fid as FileId,

    newParent as FileId,

    newName   as String    
    ) as Boolean  
    return Known(fid) and Known(newParent) 

           and then File(newParent).sort = Directory

           and newParent notin Descendants(fid)

           and newName notin { File(ch).name | ch in File(newParent).children}    

  Move(

    fid as FileId,

    newParent as FileId,

    newName   as String

    )
    require MoveEnabled(fid, newParent, newName)

    File(fid).name := newName    
    let oldParent = File(fid).parent

    if oldParent ne newParent then    
      remove fid from File(oldParent).children

      add fid to File(newParent).children

      File(fid).parent := newParent
2.4.5 Вспомогательные функции
Здесь через рекурсию определено множество «потомков».
class FileSystem

  Descendants( depth as Integer, fid   as FileId ) as Set of FileId

    if depth<0 then 

      return {}

    else
      return { fid } union 

               BigUnion({ File(ch).children | ch in Descendants(depth-1,fid) })

  Descendants( fid as FileId ) as Set of Integer

    return Descendants(Size(files), fid)
3 Моделирование работы по случайному сценарию
Одно из основных преимуществ AsmL – исполняемость. После того, как модель написана, ее можно откомпилировать и прогнать с различными значениями параметров.
3.1 Описание случайного шага
Случайный шаг состоит в том, что мы случайным образом выбираем одно из четырех действий, а затем случайно выбираем его параметры из заданного множества.
class FileSystem

  RandomCreate()

    choose parent in { file.fid | file in files },

           name in { "a", "b", "c" },

           data in { "data1", "data2" },

           sort in { Regular, Directory }

      where CreateEnabled(parent, name)

      Create(parent, name, data, sort)

  RandomUpdate()

    choose fid in { file.fid | file in files },
           data in { "data" + ToString(i) | i in { 1..100 } }

      where UpdateEnabled(fid)

      Update(fid, data)

  RandomMove()

    choose fid in { file.fid | file in files },
           newParent in { file.fid | file in files },
           newName in { "a", "b", "c" }

      where MoveEnabled(fid, newParent, newName)

      Move(fid, newParent, newName)

  RandomDelete()

    choose fid in { file.fid | file in files } 

      where DeleteEnabled(fid)

      Delete(fid)

3.2 Мониторинг 
Примитивный вывод, позволяющий отследить происходящее в системе в процессе вычисления.
class FileSystem

  ShowState()

    step forall f in files where f.fid ne root
      let pName = File(f.parent).name
      WriteLine( ToString( (f.fid, pName + "/" + f.name, f.parent, f.data) ) )

    WriteLine("")
3.3 Основной цикл
var fs = new FileSystem()

Main()

  let fs = new FileSystem()
  let actions = { "Create", "Update", "Move", "Delete" }

  step for i = 1 to 10
    step choose action in actions

      match action 

        "Create" : fs.RandomCreate()

        "Update" : fs.RandomUpdate()

        "Move"   : fs.RandomMove()

        "Delete" : fs.RandomDelete()

      WriteLine( "Step " + ToString(i) + ": Random " + action)

    step 

      fs.ShowState()    
4 Выбор тестовых последовательностей
Для построения конечного графа системы (финитизации) мы используем следующее свойство:
class FileSystem

  Structure() as Set of (FileId, FileId)

    return { (f.parent,f.fid) | f in files where f.fid ne 0 }

Кроме того, накладывается дополнительное ограничение (filter): Size(files)<5. Таким образом, помимо корневого каталога, в системе может содержаться не более 3-х файлов. После нескольких минут вычислений с указанной конфигурацией AsmL Test Tool заканчивает работу с результатом, изображенным на рисунке 1.
После  этого, можно сформировать тестовый комплект, покрывающий все ребра в данном графе. Полученный набор тестовых последовательностей можно затем сохранить в формате XML и использовать для тестирования реальных систем.

Рисунок 1
5 Заключение
Даже такое беглое ознакомление с данной технологией позволяет сформулировать некоторые открытые вопросы (направления научной работы) в данной области.
1. Является ли построенный граф полным? Строго говоря, полученный граф является оценкой снизу –  если найден пример вычисления, реализующий переход из вершины H1 в вершину H2, то это ребро включено в граф; но обратное не верно –  если в графе какие-то две вершины не соединены ребром, то это означает только то, что в процессе перебора такой пример не был найден, а не то, что его вообще не существует. В то же время вопрос полноты графа явлюется принципиальным с точки зрения качества полученного тестового комплекта.
2. Каковы стратегии построения оценки снизу для этого графа являются оптимальными? (В настоящее время в AsmL Test Tool реализовано только 2 алгоритма поиска.)
3. Пусть заданный граф финитизации построен. Как на его основе построить тестовый комплект? (Можно покрывать не только ребра, но и все простые циклы, все цепи фиксированной длины и т.д.)
4. Как выбирать свойства, на основании которых определяется финитизация? 
5. Нельзя ли автоматизировать процесс построения модели в случае когда у нас уже есть реализация системы (например, программа на С++)?

    <Sequence Id="2">


        <Step>


          <Action Name="internal Application.FileSystem.Create(System.Int32,System.String,System.String,Application.FileType) as System.Int32" IsVoid="false">


            <Instance xsi:type="xsd:string">Application.FileSystem</Instance>


            <Args>


              <anyType xsi:type="xsd:int">0</anyType>


              <anyType xsi:type="xsd:string">c</anyType>


              <anyType xsi:type="xsd:string">1</anyType>


              <anyType xsi:type="xsd:int">0</anyType>


            </Args>


          </Action>


          <Output success="true">


            <value xsi:type="xsd:int">1</value>


          </Output>


        </Step>


        <Step>


          <Action Name="internal Application.FileSystem.Move(System.Int32,System.Int32,System.String)" IsVoid="true">


            <Instance xsi:type="xsd:string">Application.FileSystem</Instance>


            <Args>


              <anyType xsi:type="xsd:int">1</anyType>


              <anyType xsi:type="xsd:int">0</anyType>


              <anyType xsi:type="xsd:string">b</anyType>


            </Args>


          </Action>


          <Output success="true" />


        </Step>

